
Abstract

The strengths of Mplus are its general modeling framework, its ease of use, its strong
customer support, and its statistical innovations that are implemented in frequent
updates of the software. Over the years, Mplus has introduced a multitude of new
analysis techniques, many of which are still available only in Mplus. This chapter gives
a summary of Mplus features and examples of some of its unique techniques.

Keywords: Structural equation modeling, item response theory (IRT), growth mod-
eling, mixture modeling, multilevel modeling, time series analysis.

1 Introduction

The strengths of Mplus are its general modeling framework, its ease of use, its strong
customer support, and its statistical innovations that are implemented in frequent
updates of the software. Over the years, Mplus has introduced a multitude of new
analysis techniques, many of which are still available only in Mplus. This chapter
gives a summary of Mplus capabilities and provides examples of some of its unique
techniques.

2 The Mplus team

Using support from a NIAAA Small Business Innovation Research (SBIR) contract,
Bengt and Linda Muthén began the development of Mplus in 1995 with the goal of
providing researchers with powerful new statistical modeling techniques. A wide gap
was seen between new statistical methods presented in statistical literature and the
statistical methods used by researchers in substantively-oriented papers. The goal was
to help bridge this gap with easy-to-use but powerful software. Version 1 of Mplus
was released in November 1998. More than 25 years later, new versions continue to be
released.

Mplus has a small team that has worked together for most of its 25 years. Linda
Muthén coordinates product development, designs the Mplus language, provides cus-
tomer support, writes the user’s guide and other materials related to Mplus. Bengt
Muthén determines statistical directions for Mplus, formulates new models, does method-
ological writing, and teaches. Tihomir Asparouhov formulates new models, algorithms,
and carries out the statistical programming. Thuy Nguyen is in charge of code control
and testing and does the interface, plotting, and R-related programming. Michelle
Conn is the office manager and handles all non-statistical aspects of the business in-
cluding interfacing with new and current users. Noah Hastings provides assistance to
all team members.

3 A quick view of Mplus capabilities

A strength of Mplus is its wide variety of analysis capabilities. The following list
of the major analysis types correspond to Chapters 3 through 12 in the Version 8
Mplus User’s Guide (Muthén & Muthén, 1998-2017) which is available at https:

//www.statmodel.com/ugexcerpts.shtml.

1. Regression and path analysis

� Linear, censored, probit, logit, multinomial, Poisson, zero-inflated Poisson,
negative binomial, zero-truncated negative binomial, and negative binomial
hurdle

2. Exploratory factor analysis

� Exploratory bi-factor, exploratory factor mixture, two-level exploratory

3. Confirmatory factor analysis and structural equation modeling

� Item response theory models (2-, 3-, 4-parameter; generalized partial credit
model), multiple-group analysis, exploratory structural equation modeling,

4. Growth modeling

� Two-part growth, discrete- and continuous-time survival analysis, N = 1 time
series analysis,

5. Mixture (latent class) modeling with cross-sectional data

� LCA, CACE, factor mixture modeling

6. Mixture modeling with longitudinal data

� Growth mixture modeling, latent class growth analysis, hidden Markov, LTA,
continuous-time survival mixture modeling

7. Multilevel modeling of complex survey data

� Clustering, stratification, and weights; two- and three-level regression, growth,
and structural equation modeling; two-level and cross-classified time series
analysis

8. Multilevel mixture modeling

� Two-level mixture regression, CFA, IRT, LCA, growth, and structural equa-
tion modeling

9. Missing data modeling

� Missing data correlates, pattern-mixture model, Diggle-Kenward selection
model, Bayesian multiple imputation of plausible values

10. Monte Carlo simulation studies

A unique feature of Mplus is that not only are these different kinds of analyses
possible, but also that the general modeling framework of Mplus makes it possible for
them to be used in combination. For example, it is possible to combine 2 with 4 by using
an exploratory factor analysis measurement model in a growth model (Asparouhov &
Muthén, 2023a; section 9); to combine 4 with 6 to do mixture continuous-time survival
analysis (Muthén et al., 2009); and to combine 4 with 9 to do growth modeling taking
into account missing data in the form of dropout (Muthén et al., 2011).

Mplus allows a wide variety of variable types. The modeling can use continu-
ous, censored, binary, ordered categorical (ordinal), unordered categorical (nominal),
counts, or combinations of these variable types. In addition, two-part (semicontinuous)
variables and time-to-event variables can be used.

The Mplus website provides inputs and data sets for the examples in the User’s
Guide (Muthén & Muthén, 1998-2017). The data for the examples are generated by
the Monte Carlo counterparts for the examples. These also provide tools for power
estimation studies and methodological research.

New developments in Mplus are listed under Version History at https://www.

statmodel.com/verhistory.shtml. Since the release of Mplus 8.0 in 2017, new fea-
tures beyond those described in the Muthén and Muthén (1998-2017) Version 8 User’s
Guide are described in the Mplus Language Addenda at https://www.statmodel.

com/ugexcerpts.shtml.
Mplus includes a dialog-based, post-processing graphics module that provides graph-

ical displays of observed data and analysis results including outliers and influential
observations. A Diagrammer can be used to draw an input model diagram, to auto-
matically create an output diagram, and to automatically create a diagram using an
Mplus input with or without analysis or data.

4 Three Examples

To illustrate Mplus analysis capabilities, three examples are discussed, their input
presented, and references for further readings listed. The first example shows the unique
strength of Mplus in structural equation modeling with categorical observed variables,
in this case for a new type of analysis of the classic MIMIC model. The second example
shows the unique strength of Mplus with categorical latent variables, that is, mixture
modeling, in this case for random intercept latent transition analysis (RI-LTA). The
third example shows the unique strength of Mplus in multilevel longitudinal modeling,
in this case for time series analysis of intensive longitudinal data (EMA data) using
dynamic structural equation modeling (DSEM).

4.1 Structural equation modeling: MIMIC

As a first example, consider the classic structural equation model called MIMIC (mul-
tiple indicators, multiple causes). For background and applications, see, e.g., Muthén
(1988, 1989) and Muthén et al. (1991). Figure 1 shows an example where y1 - y5 are
indicators of the f1, f2 factors and x1 and x2 are covariates influencing the factors.
The basic MIMIC model makes two important assumptions. First, it assumes that
the measurement structure is a confirmatory factor analysis (CFA) model so that it is
known which indicators measure which factors. In other words, the CFA specifies that
certain factor loadings are zero. In practice, however, MIMIC modeling is hindered
by the fact that a CFA specification is often not known, especially for a new measure-
ment instrument, or does not fit the data well. Second, the basic MIMIC model also
does not include direct effects from any of the covariates to any of the factor indica-
tors. The study of direct effects, however, is important because direct effects represent
measurement non-invariance with respect to the covariates. In practice, many of the
covariates are dummy variables representing different groups of individuals so that di-
rect effects capture measurement bias. Not all of these direct effects can be included
without causing a non-identified model and searching for the ones to be included can
be a cumbersome process. Mplus offers generalizations of the MIMIC model that solve
both of the above problems.

4.1.1 MIMIC with ESEM

Mplus allows relaxing the strict CFA specification of traditional MIMIC and instead
uses an exploratory factor analysis (EFA) measurement model where no zero factor
loadings need to be specified. It does so in a single analysis consisting of two steps. In
terms of Figure 1, it first estimates the Λ factor loading matrix in its unrotated form
with uncorrelated factors that have variance 1 (Ψ = I) as well as the coefficients Γ of
the regression of the factors on the covariates. In a second step, the usual EFA rotations
gives a new Λ and Ψ and the implied transformation is applied to the estimated Γ from
the first step. This is an example of exploratory structural equation modeling (ESEM)
introduced in Asparouhov and Muthén (2009). Following is an illustration with Mplus
input.

Antisocial Behavior (ASB) data were collected in the National Longtudinal Survey
of Youth (NLSY). The data set includes 17 antisocial behavior items administered in
1980 when respondents were between the ages of 16 and 23. They were asked how many

Figure 1: MIMIC model

times in the last year they engaged in: damaging property, fighting, shoplifting, stole
less than 50 dollars, stole more than 50 dollars, seriously threaten, intend to injure,
use marijuana, use other drugs, sold marijuana, sold hard drugs, con someone, take
auto, broke into building, held stolen goods. The distributions of the items are strongly
skewed and are best viewed in their dichotomous 0/1 form with 0 representing never
in the last year. A large number of background variables were also collected and 11 of
them are considered here: sex, Black, Hispanic, single, divorced, high school dropout,
college, onset of regular drinking, age, alcohol dependence, alcohol abuse. N = 7326.

Table 1 shows the Mplus input for the ESEM MIMIC analysis. General information
about Mplus statements are given in the Mplus User’s Guide (Muthén & Muthén, 1998-
2017) which is available online at https://www.statmodel.com/ugexcerpts.shtml.
The TITLE, DATA, VARIABLE commands are self explanatory. The CATEGORI-
CAL option specifies that the factor indicators are categorical, that is, either binary
or ordinal. The default is that the USEVARIABLE variables are all continuous. The
ANALYSIS command specifies that the weighted least squares estimator WLSMV of
(Muthén, 1983, 1984; Muthén & Satorra, 1995; Muthén et al., 1997) is used. This esti-
mator uses probit regression for the relationships between the factor indicators and the
factors. The bolded line of the MODEL command specifies that an EFA model should
be used with 3 factors for the 17 indicators. The keyword BY translates to measured
by. The next 3 lines specifies the regressions of the 3 factors on the 11 covariates using
the keyword ON.

Table 2 presents the ESEM factor loadings. Three factors were found to give
an interpretable solution where F1 represents property offense, F2 represents person
offense, and F3 represents drug offense. The loadings are in standardized form with
bolded values representing loadings that are significant and at least 0.2 in absolute
value. Although the three factors are clearly defined by the items in terms of large

Table 1: Mplus input for ESEM MIMIC with 17 items, 3 factors, and 11 covariates

TITLE: ESEM

DATA: FILE IS asb.dat;
FORMAT IS 34X 54F2.0;

VARIABLE: NAMES = property fight shoplift lt50 gt50 force threat
injure pot drug soldpot solddrug con auto bldg goods
gambling dsm1-dsm22 sex black hisp single divorce
dropout college onset f1 f2 f3 age94 cohort dep abuse;
USEVARIABLES = property-gambling sex black hisp
single divorce dropout college onset age94 dep abuse;
CATEGORICAL = property-gambling;

ANALYSIS: ESTIMATOR = WLSMV;

MODEL: ! 3-factor EFA:
f1-f3 BY property-gambling (*1);
f1-f3 ON sex black hisp
single divorce dropout
college onset age94 dep abuse;

OUTPUT: TECH1 TECH4 TECH10
STANDARDIZED RESIDUAL;

PLOT: TYPE = PLOT3;

Table 2: ASB factor loadings estimated by ESEM for 17 items, 3 factors, and 11 covariates
(N = 7326)

F1 F2 F3

PROPERTY 0.71 0.15 -0.04
FIGHT 0.26 0.61 -0.09
SHOPLIFT 0.61 -0.03 0.19
LT50 0.86 -0.22 0.00
GT50 0.84 -0.02 0.00
FORCE 0.39 0.33 -0.02
THREAT 0.02 0.73 0.05
INJURE -0.02 0.71 0.18
POT -0.01 -0.03 0.24
DRUG 0.01 -0.08 0.91
SOLDPOT 0.14 0.01 0.76
SOLDDRUG 0.19 0.05 0.62
CON 0.46 0.21 0.01
AUTO 0.48 0.11 0.08
BLDG 0.84 -0.01 -0.01
GOODS 0.72 0.08 0.07
GAMBLING 0.32 0.31 0.14

Correlation Matrix

F1 F2 F3

F1 1.00
F2 0.54 1.00
F3 0.61 0.25 1.00

loadings, the measurements do not have a simple CFA structure but there are several
cross loadings. The estimates are also different from EFA in that they are obtained by
including the covariates in the analysis in line with Figure 1. Each of the three factors
has significant coefficients in their regressions for 8 of the 11 covariates: sex, Black
single, divorced, dropout, onset, age, abuse.

4.1.2 MIMIC with ESEM-PSEM

As mentioned earlier, the MIMIC model does not include direct effects from covariates
to factor indicators but it is important to check if such direct effects are warranted
because they represent instances of measurement non-invariance. Allowing such direct
effects takes into account such non-invariance and allows correct estimation of both
the factor loadings and the factor regressions on the covariates. With 17 indicators
and 11 covariates, there are a total of 187 direct effects. The model including all these
direct effects is not identified and a step-wise search for which of the 187 effects are
significant is infeasible. A new approach proposed in Asparouhov and Muthén (2023a)
provides a solution to this which finds significant direct effects in a single analysis. The
solution is still little known and the example to be presented is the first of its kind.
This is the penalized structural equation modeling (PSEM) approach which makes
the non-identified model with all direct effects identified via alignment loss function
(ALF) priors. PSEM is analogous to the Bayesian BSEM approach of Muthén and
Asparouhov (2012). In both PSEM and BSEM, the priors are not based on previous
studies but are instead chosen so that the major effects from the covariates to the
factor indicators go via the factors while still allowing the data to indicate if direct
effects are present. PSEM is also akin to regularized/penalized ML where a penalty
function is added to the fit function. The precise definition, however, is custom made
for SEM and involves more than just an arbitrary penalty. PSEM is typically used with
maximum-likelihood estimation but can also be used with the weighted least squares
estimator of Mplus. The ESEM aspect of the modeling is the same as before and the
analysis is therefore referred to as ESEM-PSEM.

Table 3 shows the input for ESEM-PSEM MIMIC modeling. The top part of the
input is the same as for the ESEM input. The PSEM-specific lines are bolded. In the
MODEL command, the 187 direct effects are given parameter labels that are then given
ALF priors in the MODEL PRIORS command. The ALF(0,1) setting is standard but
Asparouhov and Muthén (2023) discusses other choices based on comparisons with a
“null model”1

The ESEM-PSEM MIMIC analysis finds 40 of the 187 direct effects to be sig-
nificant. For different values of a covariate, a direct effect corresponds to different
probabilities of endorsing the binary indicator conditional on the factors. This means
that the measurement model for this indicator is different for different values of the
covariate. For example, conditional on the factors, males are found to be more likely to
endorse the fight indicator than females and females more likely to endorse the shoplift
indicator than males. Furthermore, high school dropouts are less likely to endorse the
threat indicator, less likely to endorse the con indicator, and more likely to endorse the
building indicator.

1In this case, the null model uses an ESEM model with regressions of all indicators on all covariates but
no regressions of the factors on covariates.

9



Table 3: Mplus input excerpt for ESEM-PSEM MIMIC with 17 items, 3 factors, and 11
covariates

USEVARIABLES = property-gambling sex black hisp
single divorce dropout college onset age94 dep abuse;
CATEGORICAL = property-gambling;

ANALYSIS: ESTIMATOR = WLSMV;

MODEL: f1-f3 BY property-gambling (*1);
f1-f3 ON sex black hisp
single divorce dropout
college onset age94 dep abuse;
property-gambling ON sex-abuse (p1-p187);

MODEL PRIORS: p1-p187 ALF(0,1);

OUTPUT: TECH1 TECH4 TECH10
STANDARDIZED RESIDUAL;

PLOT: TYPE = PLOT3;

While the direct effects give important information about the measurement in-
strument, a key question is how this analysis changes the factor loadings and factor
regressions of the previous ESEM analysis. The factor loadings are largely unaffected
so that the interpretation of the factors does not change. Several of the factor regres-
sions, however, change. For the property offense factor (F1), the positive effects of
divorce and dropout covariates no longer have significant effects while the Hispanic
covariate is an added significant negative effect. For the person offense factor (F2), the
dropout covariate is an added significant positive effect. For the drug offense factor
(F3), the sex and single covariates are no longer significant.

4.1.3 Summary

The MIMIC examples show the unique strength of Mplus in the area of categorical
outcomes but the same approaches can be used with continuous outcomes. The ESEM
and PSEM approaches are Mplus innovations that offer great practical benefits. The
examples also show the strength of the weighted least squares estimation of structural
equation models which Mplus was the first software to introduce over 25 years ago.
The addition of ESEM and PSEM makes weighted least squares even more useful. The
analyses would be impractical with maximum likelihood estimation due to the model
having 3 factors and therefore 3 dimensions of numerical integration which would give
heavy computations with the large sample size of N = 7326 and would be prohibitive
with many more factors. For an overview of pros and cons of Mplus estimators with
categorical outcomes, see Muthén et al. (2015). Mplus also handles nominal and count
outcomes with continuous latent variables such as in factor analysis, including using
Bayesian analysis (Asparouhov & Muthén, 2021). The PSEM approach has a wide
variety of application beyond what has been discussed here. It enables many types of
new factor analysis models as well as exploratory growth models.

Examples of other unique Mplus features with structural equation modeling in-
clude random intercept cross-lagged panel modeling (RI-CLPM; Hamaker et al., 2015)
extended to binary and ordinal variables (Muthén et al., 2023) and causal inference
with counterfactually-defined direct and indirect effects with categorical mediators and
outcomes (Muthén et al., 2016; Muthén & Asparouhov, 2015). Mplus also offers a rich
set of models for item response theory (IRT) applications (Muthén & Asparouhov,
2016) with combinations of continuous, binary, ordinal, nominal, and count outcomes,
including general partial credit models and 3- and 4-parameter item response models.

The MIMIC examples covered the combination of categorical observed and con-
tinuous latent variables. The next example concerns another unique Mplus capability
where the observed variables are still categorical but where there are both continuous
and categorical latent variables, the latter resulting in mixture modeling with latent
classes.

4.2 Mixture modeling: Latent transition analysis (LTA,
RI-LTA)

Figure 2 shows an example of a latent transition model for 3 timepoints where at each
time point, 2 indicators are measured. The categorical latent variables are referred to
as c1, c2, c3. The model has four components. First, there is a measurement model at
each timepoint t relating the latent class indicators ut to the latent class variables ct

Figure 2: RI-LTA for two latent class indicators at three timepoints

u11

c1 c2 c3

u12 u31 u32u21 u22

f

λ1 λ1 λ1 λ2λ2λ2

via the conditional probabilities P (ut|ct). This gives a latent class measurement model
for each timepoint where the typical assumption is measurement invariance over time.
Second, there are also initial status probabilities P (c1) describing the distribution of
the latent classes of c1. Third, there are transition probabilities P (c2|c1) and P (c3|c2)
which are the key parameters of the model. A fourth component is the factor f at the
top of the figure. This latent continuous variable is referred to as a random intercept
factor and is a novel feature introduced in Muthén and Asparouhov (2022) under the
name random intercept latent transition analysis (RI-LTA). A detailed description of
modeling issues and Mplus analysis using LTA and RI-LTA is given in Muthén (2023a).

The need for a random intercept is common in multilevel modeling and latent
transition analysis is multilevel due to having repeated measurements over time within
individuals. A random intercept captures “unobserved heterogeneity”, in this case
variation across individuals that is constant over time. In psychological analyses the
random intercept is referred to as a stable trait. In Figure 2, each of the two latent
class indicators for a given timepoint is influenced by the random intercept factor f
with factor loadings λ. The model has factor mean zero and fixed unit factor variance.
Alternatively, the factor variance is estimated and the factor loadings fixed at 1 as
is common with random intercepts. With different latent class indicators, different
factor loadings reflect different influence of the stable trait on different measurements.
To reflect the time invariant influence of the stable trait, the loading for each latent
class indicator is held equal across time. The introduction of a random intercept
factor solves the problem of regular LTA where between- and within-person sources of
variation are confounded. In RI-LTA, the transition probabilities for the latent class
variables represent a within-person process free of between-person differences. This is
analogous to the random intercept version of cross-lagged panel modeling, RI-CLPM.

The illustration of LTA and RI-LTA draws on data from a reading proficiency study
in Kindergarten and first grade (Kaplan, 2008) with 4 time points: Fall and Spring of
Kindergarten and Fall and Spring of Grade 1. N = 3574. The outcomes are 5 binary

items representing mastery of:

� Basic reading skills of letter recognition

� Beginning sounds

� Ending sounds

� Sight words

� Words in context

Kaplan (2008) hypothesized that three latent classes underly the 5 outcomes at each
time point with the idea that these latent classes correspond to three stages of learning:
Low alphabet knowledge, early word reading, early reading comprehension.

The Mplus input for RI-LTA is shown in Table 4. The bolded line of CLASSES
specifies that 4 latent class variables c1-c4 each have 3 latent classes. In the ANALYSIS
command, the bolded first line specifies mixture modeling. The latent class indicators
are related to the latent class variables via logistic regressions. Maximum likelihood
estimation is the default for mixture modeling. To find the best solution, multiple
sets of starting values are required using the STARTS option. Because of the random
intercept factor, this analysis requires the option ALGORITHM = INTEGRATION. In
the Overall part of the MODEL command, the bolded lines refer to the specification
of the random intercept factor measured by the 5 latent class indicators at the 4
timepoints with loadings held equal over time by the parameter labels (p1-p5). The
factor mean is fixed at zero and the factor variance is fixed at 1. Regular LTA does not
include these bolded lines. The c ON c lines specifies lag 1 transitions between the 4
latent class variables. Table 5 continues the input with the latent class specific model
statements. Here, the logit thresholds for the latent class indicators are specified as
varying across the latent classes but time invariant, that is, held equal across the 4
latent class variables.

For this example, the regular LTA model fits substantially worse than the RI-
LTA. For the hypothesized 3-class model, LTA has 35 parameters with a maximum
loglikelihood value of -21,793 with BIC = 43,873. RI-LTA has 5 more parameters
due to the 5 loadings on the random intercept factor and gets a better loglikelihood
value of -20,329 with a considerably better (lower) BIC = 40,984. The better fit of
RI-LTA is a typical finding as evidenced by the examples used in Muthén (2020). LTA
is unnecessarily restrictive and gives distorted results.

Table 6 shows estimated latent transition tables. Two of the three transitions
are presented, Fall of Kindergarten to Spring of Kindergarten and Fall of 1st grade to
Spring of 1st grade. The top part of the table shows estimates from regular LTA and the
bottom part shows estimates from RI-LTA. Comparing the top part with the bottom
part for each transition table shows that RI-LTA presents a quite different picture of
the development of reading proficiency than LTA. For example, in the transition table
on the left showing transitions from Fall of Kindergarten to Spring of Kindergarten,
RI-LTA says that there is only a 0.17 probability of staying in the lowest class of low
alphabet knowledge and a 0.82 probability of moving up to the middle class of early
word reading. In contrast, LTA says that the probability of staying is higher, 0.34, and
the probability of moving is lower, 0.65. In the transition table to the right showing
transitions from Fall 1st to Spring 1st, RI-LTA says that the probability of staying in
the middle class of early word reading is 0.02 and that the probability of moving up

Table 4: Mplus input for RI-LTA of reading proficiency

TITLE: RI-LTA

DATA: FILE = dp.analytic.dat;
FORMAT = f1.0, 20f2.0;

VARIABLE: NAMES = pov
letrec1 begin1 ending1 sight1 wic1
letrec2 begin2 ending2 sight2 wic2
letrec3 begin3 ending3 sight3 wic3
letrec4 begin4 ending4 sight4 wic4;

USEVARIABLES: letrec1 begin1 ending1 sight1 wic1
letrec2 begin2 ending2 sight2 wic2
letrec3 begin3 ending3 sight3 wic3
letrec4 begin4 ending4 sight4 wic4;

CATEGORICAL: letrec1 begin1 ending1 sight1 wic1
letrec2 begin2 ending2 sight2 wic2
letrec3 begin3 ending3 sight3 wic3
letrec4 begin4 ending4 sight4 wic4;

CLASSES: c1(3) c2(3) c3(3) c4(3);

MISSING = .;

ANALYSIS: TYPE = MIXTURE;
STARTS = 80 16;
PROCESSORS = 8;
ALGORITHM = INTEGRATION;
! INTEGRATION = 30; ! Default = 15

MODEL: %OVERALL%
f BY letrec1-wic1* (p1-p5)
letrec2-wic2* (p1-p5)
letrec3-wic3* (p1-p5)
letrec4-wic4* (p1-p5);
[f@0]; f@1;

c2 ON c1;
c3 ON c2;
c4 ON c3;
! Input continues on next page

Table 5: Mplus input for RI-LTA, continued

MODEL c1: %c1#1%
[letrec1$1-wic1$1] (1-5);
%c1#2%
[letrec1$1-wic1$1] (6-10);
%c1#3%
[letrec1$1-wic1$1] (11-15);

MODEL c2: %c2#1%
[letrec2$1-wic2$1] (1-5);
%c2#2%
[letrec2$1-wic2$1] (6-10);
%c2#3%
[letrec2$1-wic2$1] (11-15);

MODEL c3: %c3#1%
[letrec3$1-wic3$1] (1-5);
%c3#2%
[letrec3$1-wic3$1] (6-10);
%c3#3%
[letrec3$1-wic3$1] (11-15);

MODEL c4: %c4#1%
[letrec4$1-wic4$1] (1-5);
%c4#2%
[letrec4$1-wic4$1] (6-10);
%c4#3%
[letrec4$1-wic4$1] (11-15);

OUTPUT: TECH10 TECH15;

Table 6: Estimated latent transition tables for LTA versus RI-LTA

Regular LTA Regular LTA

1 2 3 1 2 3

Spring K Spring 1st

1 0.34 0.65 0.01 1 0.26 0.51 0.23
Fall K 2 0.00 0.65 0.35 Fall 1st 2 0.01 0.13 0.86

3 0.000 0.000 1.00 3 0.00 0.000 1.00

RI-LTA RI-LTA

1 2 3 1 2 3

Spring K Spring 1st

1 0.17 0.82 0.01 1 0.16 0.00 0.84
Fall K 2 0.000 0.82 0.19 Fall 1st 2 0.00 0.02 0.98

3 0.000 0.000 1.00 3 0.01 0.00 0.99

to the early reading comprehension class is 0.98. The corresponding probabilities for
LTA are 0.13 and 0.86.

4.2.1 Summary

The RI-LTA example features a combination of observed and latent variables where
the latent variables are both continuous and categorical (latent class variables), lead-
ing to general types of mixture modeling. Since its launch 25 years ago, Mplus has
offered a unique combination of mixture modeling, structural equation modeling, and
multilevel modeling. There are several other notable examples with this combina-
tion. Muthén and Asparouhov (2009a) discussed multilevel regression mixture analy-
sis, Henry and Muthén (2010) presented multilevel latent class analysis examples with
latent class variables on two different levels, Asparouhov and Muthén (2008, 2016) dis-
cussed general multilevel mixture models, and Muthén and Shedden (1999), Muthén
(2002), Muthén and Brown (2009), Muthén and Asparouhov (2009b), Muthén et al.
(2002) presented general growth mixture modeling. The Mplus User’s Guide presents
several more unique mixture modeling examples including Complier Average Causal
Effect (CACE) modeling, discrete-time survival mixture analysis, and continuous-time
survival mixture analysis.

4.3 Multilevel time series analysis: Dynamic structural
equation modeling (DSEM)

Figure 3 shows a dynamic structural equation model for two outcomes, y and z. As
usual, squares represent observed variables. For simplicity, only two time points are
shown, t and t − 1, but this gives the essence of the model. The observed variables
are decomposed into between and within individual components of latent variables,
represented by the circles on those two levels. For example, the figure shows that yt
is decomposed into the sum of yWt on within and yB on between. Each of the two
levels has a model and the model can be a multivariate structural equation model. For
example, a mediation model or a factor analysis model can be analyzed. The model is a
multilevel time series model where the within part of the model corresponds to a vector
auto-regressive (VAR) model in time series terms. It is a time series generalization of
a cross-lagged panel model. A unique feature of Mplus DSEM is the contemporaneous
effect of z on y. In time series analysis, a single individual is typically considered
(N = 1) but in social science applications it is more common to analyze a group of
individuals (N > 1). Because of this, the parameters on the within level can be random,
that is, varying across individuals. Mplus is unique in allowing a variety of random
effects including random residual variances and covariances (not shown in the figure).
The random effects are marked by filled circles on the within level. All of the random
effects have a latent variable counterpart on the between level and in the figure these
are regressed on a between-level covariate w. The full model in the figure is a time
series generalization of RI-CLPM that includes a contemporanous effect and lets all
parameters be random so that the within model is specific to each individual. To
handle the estimation of the general model, Bayesian analysis is used. DSEM theory
is given in Asparouhov et al. (2018) and Asparouhov & Muthén (2019, 2020). For a
detailed description of how to use Mplus to do DSEM, see Muthén and Asparouhov
(2023).

Data for the DSEM illustration is from a study designed to detect at-risk mood
profiles related to depression in adolescents (see, e.g., de Haan-Rietdijk et al., 2017 and
Dietvorst et al., 2021).2 ESM questionnaires measuring positive and negative affect
were administered to 240 Dutch adolescents ages 12 to 16 with 63% girls. Positive
affect (PA) was measured as the average of six 7-category items. Several measures per
day were collected for seven days, Tuesday - Monday, including the covariates gender,
age, SDQ (measure of childhood emotional problems), and tiredness. Participants
filled out ESM questionnaires at random time throughout the day, including during
school hours with questionnaires delivered on the adolescents’ own smartphones. The
individually-varying random time points are handled as described in Hamaker et al.
(2023) and Muthén and Asparouhov (2023), synchronizing time by inserting missing
data for individuals when times are not observed. A choice is made to represent the
24 hours by 12 2-hour intervals. Having 12 measurements per day results in T = 84 as
the maximum number of timepoints.

Table 7 shows the Mplus input for two-level DSEM analysis of PA and Tired using
the VAR model with an added contemporaneous effect. The CLUSTER statement
specifies that clusters are individuals represented by the id variable. The TINTERVAL
statement specifies that the hrs variable is the recorded time, that a time interval

2Thanks are due to Loes Keijsers, PI, who provided the data.

Figure 3: Two-level DSEM with cross-lagged, contemporaneous, and random effects

Table 7: Mplus input excerpt for two-level DSEM VAR model with a contemporaneous effect

USEVARIABLES = pa tired;
CLUSTER = id;
TINTERVAL = hrs (2 time);
LAGGED = pa(1) tired(1);

Analysis: TYPE = TWOLEVEL;
ESTIMATOR = BAYES;
BITERATIONS = (2000);
PROCESSORS = 2;

MODEL: %WITHIN%
pa ON pa&1 tired tired&1;
tired ON tired&1 pa&1;

%BETWEEN%
pa WITH tired;

OUTPUT: STANDARDIZED TECH1 TECH4 TECH8;

PLOT: TYPE = PLOT3;

of 2 hours is used, and that the new time variable with inserted missing data lines
is called time. The LAGGED statement specifies that a lag 1 model is used for PA
and Tired. In the ANALYSIS command, the Bayes estimator is requested with a
minimum of 2000 iterations. In the MODEL command, the within-level use of pa&1
and tired&1 on the right-hand side of ON on specifies that pa and tired with lag 1 are
predictors. Here, pa ON pa&1 refers to the auto-regression of PA. In contrast, tired on
the right-hand side of ON specifies that tired with lag 0 is a predictor, representing the
contemporaneous effect of tired on PA. The betwen level simply specifies a covariance
between the between-level components of PA and tired, also estimating their between-
level variances.

The standardized regression coefficients for PA regressed on tired with lag 0 and lag
1 are -0.19 and 0.00, respectivey where only the lag 0 effect is significant. This shows
the need for allowing a contemporanous effect. As for the reverse effect, the coefficient
for tired regressed on PA with lag 1 is insignificant with a standardized estimate of
only -0.01.

Table 8 shows a model variation referred to as residual DSEM (RDSEM). Based
on the results of the previous analysis, there are no cross-lagged effects and the tired
variable is viewed as a contemporaneous covariate for PA. There is not a direct effect
from PAt−1 to PAt but instead an auto-regression among the residuals for PA in the
regressions on tired which is more in line with regular two-level modeling. RDSEM
uses the hat (ˆ) notation for this residual specification. The contemporaneous effect

Table 8: Mplus input excerpt for RDSEM with a random slope for a contemporaneous effect

USEVARIABLES = pa tired age SDQ girl;

CLUSTER = id;
TINTERVAL = hrs (2 time);
BETWEEN = age SDQ girl;
LAGGED = pa(1) tired(1);

ANALYSIS: TYPE = TWOLEVEL RANDOM;
ESTIMATOR = BAYES;
BITERATIONS = (2000);
PROCESSORS = 2;

MODEL: %WITHIN%
paˆ ON paˆ1;
s | pa ON tired;
tiredˆ ON tiredˆ1;

%BETWEEN%
pa tired s ON age SDQ girl;

OUTPUT: STANDARDIZED TECH1 TECH4 TECH8;

PLOT: TYPE = PLOT3;

of tired on PA is now specified as random slope using the bar (|) notation, s | pa ON
tired. On the between level, the random slope s and the between-level parts of PA and
tired are regressed on the covariates age, SDQ, and girl.

For the regression of PA on tired, the within-level standardized estimate averaged
over individuals is about the same as before, -0.20. The random slope s has a significant
regression on SDQ with coefficient -0.15 so that more childhood problems lowers the
effect of tired on PA. The between-level component of PA also has a significant negative
relation to SDQ, -0.34, so that higher SDQ both lowers the level of PA over time and
lowers the effect of tired. The between-level component of tired has a significant positive
relation to SDQ, 0.22.

Table 9 shows a final analysis variation using a factor analysis model. Here, the
6 items that the PA score is comprised of are specified to measure two factors where
the factors correspond to a low- and high-arousal factor structure found in Muthén
et al. (2024). The two factors are specified using the BY statements on the within
level and on the between level. The (&1) notation on the within level means that the
factors can be modeled as lag 1. The two factors are regressed on tired using random
slopes and the RDSEM approach specifies auto-regressions among the residuals of those
regressions. On the between level, the factors and the random slopes are regressed on

Table 9: Mplus input excerpt for RDSEM factor analysis with contemporaneous random
slopes

CLUSTER = id;
TINTERVAL = hrs (2 time);
LAGGED = tired(1);
BETWEEN = age SDQ girl;

ANALYSIS: TYPE = TWOLEVEL RANDOM;
ESTIMATOR = BAYES;
BITERATIONS = (1000);
PROCESSORS = 2;

MODEL: %WITHIN%
fw1 BY pala1-pala3* (&1);
fw2 BY paha3* paha2 paha1(&1);
fw1-fw2@1;
fw1ˆ ON fw1ˆ1;
fw2ˆ ON fw2ˆ1;
s1 | fw1 ON tired;
s2 | fw2 ON tired;
tiredˆ ON tiredˆ1;

%BETWEEN%
fb1 BY pala1-pala3*;
fb2 BY paha3* paha2 paha1;
fb1-fb2@1;
fb1-fb2 s1 s2 ON tired age SDQ girl;

the between-level part of tired and the three between-level covariates.
The results show that the within-level standardized estimate averaged over individ-

uals for the within-factor regression slope on tired is much larger for the high-arousal
factor than for the low-arousal factor, -0.30 versus -0.07. On the between level, the
between-level parts of the factors both have significant negative slopes for tired and
SDQ. The between-level results also show that the random slope for the low-arousal
factor has significant negative slopes for tired and SDQ, while the random slope for
the high-arousal factor has a significant negative slope only for tired.

4.3.1 Summary

The DSEM examples show only a small subset of the unique capabilities of multilevel
time series analysis in Mplus. It is also possible to add a third level with a specification
of across-time variation useful for detecting trends in the data, referred to as cross-
classified analysis. In addition to capturing growth modeling features, this enables
a flexible analysis of cycles in the data (Muthén et al., 2024a). Floor effects and
trend analysis can also be handled (Muthén et al., 2024b). Continuous-time DSEM
(Asparouhov & Muthén, 2024) avoids the discretization of time and can be used to find
optimal time intervals that maximize the effects in lagged relationships. More types of
analyses are presented in Asparouhov et al. (2018), Hamaker et al. (2023) with further
details in the web talk of Muthén and Asparouhov (2023). A list of applications is given
on the website page https://www.statmodel.com/TimeSeries.shtml.

5 Other Mplus features

Many unique analysis possibilities have been left out in this brief summary of Mplus
capabilities. Examples include measurement invariance alignment studies of multiple
groups (Asparouhov & Muthén, 2023b) as well as multiple timepoints (Muthén, 2023b).
A fuller list is provided on the Mplus home page at https://www.statmodel.com/

index.shtml. The left column shows links to Special Mplus Topics as well as Mplus
Web Talks with specific discussion of inputs. The right column shows a link to Papers
using special Mplus features. The right column also shows links to Recent Papers and
Mplus Short Courses.

